

CASE STUDIES IN OPTIMIZATION: CATENARY PROBLEM

IGOR A. GRIVA AND ROBERT J. VANDERBEI

Operations Research and Financial Engineering
Princeton University
ORFE

Revised June 25, 2003

ABSTRACT. This is the second paper in a series presenting case studies in modern large-scale constrained optimization [9], the purpose of which is to illustrate how recent advances in algorithms and modeling languages have made it easy to solve difficult problems using modern optimization software. In this paper, we consider the shape of a hanging chain, which, in equilibrium, minimizes the potential energy of the chain. In addition to the tutorial aspects of this paper, we also emphasize the importance of certain modeling issues such as convex vs. nonconvex formulations of given problem. We will present several models of the problem and demonstrate differences in the number of iterations and solution time.

1. INTRODUCTION

The hanging chain or catenary problem (the word “catenary” comes from the Latin word “catena” meaning chain) was first posed in the *Acta Eruditorum* in May 1690 by Jacob Bernoulli as follows: “To find the curve assumed by a loose string hung freely from two fixed points”. Earlier, Galileo mistakenly conjectured that the curve was a parabola. Later Joachim Jung proved that the curve cannot be a parabola but without presenting any solution of the real curve. In June 1691 there were three solutions published, from Leibniz, Huygens and Johann Bernoulli brother of Jacob. Even though these mathematicians approached this problem in three different ways they concluded that the curve was the hyperbolic cosine, which then came to be known as the catenary.

In more recent times, the catenary curve has come to play an important role in civil engineering. The solution of the catenary problem provides the starting point for consideration of the effects on a suspended cable of extraneous applied forces such as arising from the live loads on a practical suspension bridge.

However, in the real world, the problem of finding an optimal construction shape is more complicated than the original catenary problem. Jacob Bernoulli assumed that the string is flexible and of uniform cross-section, which implies that every segment of equal length has equal mass. This assumption is too

Date: June 25, 2003.

Key words and phrases. shape optimization, optimal design, constrained optimization.

restrictive for modern engineering. Moreover, in many practical suspension bridges the total weight of the bridge, instead of being uniformly distributed along the cable, is actually more uniformly distributed across the bridge span. In this case, the shape of the cable is closer to a parabola than a catenary.

The purpose of this paper is to demonstrate how a general chain problem can be modeled as an optimization problem and solved by widely available optimization tools. Throughout the paper we present several optimization models and demonstrate how to solve them using optimization software. We express these models in the AMPL modeling language [5]. This language is used as a common mechanism for conveying optimization problems. We selected LOQO [6, 7, 8] as a nonlinear optimization solver, which implements an interior point method (IPM) for general nonlinear optimization and adequately serves our needs.

This paper is intended to be a tutorial on optimization. We emphasize the importance of proper modeling.

The main point we wish to illustrate is that any given physical problem can have multiple equivalent mathematical formulations some of which are numerically tractable while others are not. More specifically, we focus on convex vs. nonconvex formulations of a problem. An optimization problem is convex if it can be expressed as:

$$\min f(x),$$

subject to

$$c_i(x) \leq 0, \quad i = 1, \dots, m,$$

$$a_j(x) = 0, \quad j = 1, \dots, p,$$

where $f(x)$, $c_i(x)$, $i = 1, \dots, m$ are convex and $a_j(x)$, $j = 1, \dots, p$ are affine functions. Convex optimization problems are much more numerically tractable than nonconvex ones.

The paper is organized as follows. In the next section, we consider the catenary problem with the mass uniformly distributed along its length. We consider two modeling approaches. The first one leads to the solution of the nonconvex formulation of the catenary problem in the direct form $y(x)$, $x \in [x_a, x_b]$. The second model leads to the solution of the convex formulation of the problem in a parametric form $(x(t), y(t))$, $t \in [0, l]$.

Section 3 considers the case when the mass of a chain is uniformly distributed along the horizon. Again, as in the previous section, two modeling approaches are considered leading to $y(x)$ and $(x(t), y(t))$ respectively.

2. CHAIN PROBLEM, DISCRETIZATION ALONG THE HORIZON.

The common formulation of the variational problem for finding the shape of a heavy chain of length l with uniformly distributed mass along the chain is to find a function $y(x)$ that minimizes the potential energy

$$(1) \quad \min \int_{x_a}^{x_b} y \sqrt{1 + y'^2} dx,$$




FIGURE 1. Catenary

where y is height and $\sqrt{1 + y'^2}dx$ is proportional to mass, subject to constraints:

$$(2) \quad \int_{x_a}^{x_b} \sqrt{1 + y'^2} dx = l,$$

and

$$(3) \quad y(x_a) = y_a, \quad y(x_b) = y_b.$$

The solution to this problem is

$$(4) \quad y(x) = C \cosh \frac{x + C_1}{C} + C_2,$$

where the values of C , C_1 and C_2 are determined by conditions (2)-(3) ([1, 2]). Figure 1 shows the graphical representation of $y(x)$.

2.1. Discretization. In [9], we considered two simple methods for discretizing calculus of variations problems: the trapezoidal discretization and the midpoint discretization. We showed that the trapezoidal method, which seems to be the most widely used method, is more likely to exhibit non uniqueness of the solution when there are extra degrees of freedom over which one is optimizing. In this paper, we shall use the midpoint discretization in all of the examples.

If $x_a = x_0 < x_1 < \dots < x_{N-1} < x_N = x_b$ is the uniform discretization of segment $[x_a, x_b]$, the midpoint method approximates integrals as follows

$$\int_a^b f(x) dx \approx \sum_{i=0}^{N-1} \frac{f(x_{i+\frac{1}{2}})}{(x_{i+1} - x_i)}, \quad \text{where} \quad x_{i+\frac{1}{2}} = \frac{x_i + x_{i+1}}{2}.$$

<pre> param N := 200; param d := 5; param ax := 0; param ay := 0; param bx := d; param by := 0; param dx := (bx - ax)/N; param x {j in 0..N} := ax*(1-j/N) + bx*j/N; param l := 2*d; param g := 9.8; param m {0..N}, default 1; var y {0..N}; var ydot {j in 1..N}; minimize energy: sum {j in 1..N} dx*m[j]*g*sqrt(1+ydot[j]^2)* (y[j]+y[j-1])/2; </pre>	<pre> s.t. eqn {j in 1..N}: y[j] = y[j-1] + dx*ydot[j]; s.t. length: sum {j in 1..N} sqrt(1+ydot[j]^2)*dx = l; s.t. leftfixed: y[0] = ay; s.t. rightfixed: y[N] = by; let {j in 0..N} y[j] := (j/N)*by + (1-j/N)*ay; let {j in 1..N} ydot[j] := (y[j]-y[j-1])/dx; solve; </pre>
---	--

FIGURE 2. The AMPL model mid.mod

Also, the method approximates the derivatives as follows

$$y'(x_{j+\frac{1}{2}}) = \frac{y_{j+1} - y_j}{x_{j+1} - x_j}, \quad j = 0, 1, \dots, N-1$$

We avoid creating extra variables $y(x_{j+\frac{1}{2}})$ in the middle of discretization intervals by approximating $y(x_{j+\frac{1}{2}}) \approx 0.5(y(x_{j+1}) + y(x_j))$. The function used to approximate integral (1) is $f(x_{i+\frac{1}{2}}) = 0.5[y(x_{j+1}) + y(x_j)]\sqrt{1 + y'(x_{j+\frac{1}{2}})^2}$. The midpoint discretization of problem (1)-(3) in the AMPL modeling language is shown in Figure 2.

We solved this problem with LOQO for various values of discretization parameter N . Table 1 shows the behavior of the solver.

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	1.41	14.68	248.02	
pr. val =	-223.2416282	-223.2449622	-223.2457957	
dual val =	-223.2416282	-223.2449622	-223.2457957	
pr. inf =	6.1e-12	1.6e-11	1.8e-12	
dual inf =	2.3e-11	4.9e-11	1.0e-11	
steps	50	90	199	≥ 500

TABLE 1. Chain model, discretized along the horizon

2.2. Catenary problem – discretization along the chain. The discretized constraint (2) can be a source of numerical problems for an interior point algorithm. We found that parameterization of the chain along itself can lead to a model that is simpler to attack both mathematically and numerically.




FIGURE 3. Catenary

Let's consider again a chain with uniformly distributed mass along the cable. We look for $(x(t), y(t))$, $t \in [0, l]$, which minimizes the potential energy

$$(5) \quad \min \int_0^l m(t)y(t)dt,$$

such that

$$(6) \quad \dot{x}^2 + \dot{y}^2 = 1,$$

and

$$(7) \quad x(0) = x_a, \quad y(0) = y_a, \quad x(l) = x_b, \quad y(l) = y_b.$$

where $m(t) = m = \text{const}$, $\dot{x} = dx/dt$, $\dot{y} = dy/dt$. The following proposition shows that the solution of problem (5)-(7) coincides with that of (1)-(3).

Proposition 1. *Let $(x(t), y(t))$, $t \in [0, l]$ be the solution of problems (5)-(7). The function $x(t)$ is strictly monotone and hence invertible. Furthermore,*

$$y(t^{-1}(x)) = C \cosh \frac{x - C_1}{C} + C_2,$$

where the values of C , C_1 and C_2 are determined by conditions (6)-(7).

Proof. We can assume that $m = 1$. Also, we can assume that $\dot{x} \neq 0$, $t \in [0, l]$. Otherwise it would contradict the laws of physics: the horizontal components of tensions T_1 and T applied to any portion of the chain SR should be equilibrated (see Figure 3). Therefore if $(x(t), y(t))$ is the solution of problem (5)-

(7) there exists a multiplier $\lambda = \lambda(t)$, $t \in [0, l]$ such that $(x(t), y(t))$ is an extremal without side condition for the functional (see [1])

$$(8) \quad \min \int_0^l (y + \lambda(\dot{x}^2 + \dot{y}^2 - 1)) dt$$

The Euler equations for problem (8) are

$$\begin{aligned} \frac{d}{dt} 2\lambda\dot{x} &= 0, \\ \frac{d}{dt} 2\lambda\dot{y} &= 1. \end{aligned}$$

Integrating the first equation we obtain $2\lambda\dot{x} = C$. Keeping in mind that $\dot{x} \neq 0$, $t \in [0, l]$ we can express $\lambda = C/(2\dot{x})$ and substitute into the second equation, which becomes

$$(9) \quad C \frac{d}{dt} \frac{\dot{y}}{\dot{x}} = 1.$$

Since we look for a solution of the form $y(x)$, we have

$$(10) \quad C \frac{d}{dt} \frac{\dot{y}}{\dot{x}} = C \frac{d}{dt} y' = C y'' \dot{x}$$

(Note: we use “dot” to denote differentiation with respect to t and “prime” to denote differentiation with respect to x). Also (6) turns to

$$\dot{x}^2 + \dot{y}^2 = \dot{x}^2 + (y'\dot{x})^2 = 1,$$

which is equivalent to

$$(11) \quad \dot{x} = \frac{1}{\sqrt{1 + y'^2}},$$

under the assumption that $\dot{x} > 0$, $t \in [0, l]$. Therefore combining (9), (10), and (11) we obtain the following ODE

$$\frac{C y''}{\sqrt{1 + y'^2}} = 1,$$

which can be rewritten as

$$C \frac{dy'}{\sqrt{1 + y'^2}} = dx.$$

After integrating this ODE we have

$$x = C \sinh^{-1} y' + C_1,$$

which is equivalent to

$$y' = \sinh \left(\frac{x - C_1}{C} \right).$$

Finally after another integration we obtain

$$y(x) = C \cosh \frac{x - C_1}{C} + C_2.$$

This proves the proposition.

Figure 4 presents the discretized problem (5)-(7). This optimization problem has N constraints:

<pre> param N := 200; param d:=5; param ax := 0; param ay := 0; param bx := d; param by := 0; param l :=2*d; param g := 9.8; param m {1..N-1}, default 1; var x {0..N}; var y {0..N}; minimize energy: (l/N)*sum {j in 1..N-1} m[j]*g*y[j]; </pre>	<pre> s.t. link {j in 1..N}: (x[j]-x[j-1])^2+(y[j]-y[j-1])^2 = (l/N)^2; s.t. xleftfixed: x[0] = ax; s.t. yleftfixed: y[0] = ay; s.t. xrightfixed: x[N] = bx; s.t. yrightfixed: y[N] = by; let {j in 0..N} x[j] := (j/N)*b_x + (1-j/N)*a_x; let {j in 0..N} y[j] := (j/N)*b_y + (1-j/N)*a_y; solve; </pre>
--	---

FIGURE 4. The AMPL model chain.mod

$$\text{s.t. link } \{j \text{ in } 1..N\}: (x[j]-x[j-1])^2+(y[j]-y[j-1])^2 = (l/N)^2;$$

which are a potential source of numerical problems. Indeed, LOQO fails to solve this problem even with $N = 100$.

However, the formulation can be simplified drastically by relaxing the equality in the “link” constraints into inequalities:

$$\text{s.t. link } \{j \text{ in } 1..N\}: (x[j]-x[j-1])^2+(y[j]-y[j-1])^2 \leq (l/N)^2;$$

It can be shown by contradiction that an optimal solution to the problem in Figure 4 with the relaxed “link” constraints satisfies the constraints as equalities. Indeed, suppose that one or more “link” constraints are strict inequalities for an optimal solution. Then by rotating sub chains around fixed points A and B (see Figure 1) or by lowering the sub chain between multiple inequality links we can decrease the potential energy of the whole chain, which contradicts optimality.

At the same time, the relaxed problem is convex with a linear objective function and easily solvable by any nonlinear optimization code in particular LOQO. Table 2 displays solution statistics.

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	0.12	0.24	0.53	1.10
pr. val =	-223.2381141	-223.2440837	-223.2455761	-223.2459492
dual val =	-223.2381140	-223.2440837	-223.2455761	-223.2459492
pr. inf =	1.4e-11	3.0e-12	2.6e-13	2.6e-13
dual inf =	2.7e-12	9.4e-13	1.5e-14	1.2e-14
steps	22	24	27	28

TABLE 2. Parameterized chain

If the mass function is not a constant along the chain but is instead any general known function $m(t)$ then it is still easy to obtain a solution of a hanging chain problem. For example, let the mass of each node equal 1 except for three special nodes: the center node and the two nodes one quarter of the length away from both end points. This specification is modeled by making the following addition to the AMPL code in Figure 4:




FIGURE 5. Not uniformly distributed mass function

```
let {j in 1..3} m[int((N-1)*j/4)] := 0.3*N;
```

The resulting shape of the chain is shown in Figure 5 while Table 3 presents the statistics of the solver's behavior.

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	0.21	0.35	0.80	1.80
pr. val =	-474.7789365	-479.2553063	-481.4659418	-482.5641514
dual val =	-474.7789365	-479.2553063	-481.4659417	-482.5641513
pr. inf =	8.0e-12	3.4e-12	7.5e-12	4.0e-12
dual inf =	1.1e-11	7.8e-12	3.0e-11	2.4e-11
steps	36	36	42	46

TABLE 3. Not uniformly distributed mass function

2.3. Lessons. Hanging chain problems are convex and therefore easy to solve if the model representation is also convex. It is apparently not possible to make a convex model when parameterizing along x . But it is easy to do so when parameterizing along the chain. Indeed, the convexity is created when constraint (5) is relaxed to the inequality without changing the optimal solution. However, if the model is discretized along the horizon, the relaxation of constraint (2) to inequality gives a model that is not equivalent and results in an incorrect solution (see Figure 6). The length of an extremal chain turns out to be less than l , and thus we cannot employ this trick for problem (1)-(3). Hence, the parameterized model (5)-(7) is more suitable for the computational analysis.




FIGURE 6. Relaxed model, discretized along the horizon

3. BRIDGE PROBLEM.

The hanging chain model can be the starting point for studying other problems of structural design in civil engineering. For example, let's consider the problem of identifying the shape of the cable that carries a suspension bridge. In this case, the total weight of the bridge, instead of being uniformly distributed along the cable, is actually more uniformly distributed across the bridge span. This weight distribution makes the bridge model more difficult to analyze.

In this section, we consider the bridge model and show how it can be solved using LOQO.

3.1. Bridge problem, discretization along the horizon. Let us first try to build a bridge model discretized along the horizontal axis, which implies that the mass function is constant: $m(x) = m$ for any x . The variational problem that reflects this situation could be formulated as follows:

$$(12) \quad \min \int_{x_a}^{x_b} y(x) dx,$$

such that

$$(13) \quad \int_{x_a}^{x_b} \sqrt{1 + y'^2} dx = l,$$

with

$$(14) \quad y(x_a) = y_a, \quad y(x_b) = y_b.$$

Surprisingly, the solution $y(x)$ to this problem is an arc segment of a circle:

Proposition 2. *If $0 < l_1 \leq l \leq l_2$, for some constants l_1 and l_2 then the solution $y(x)$, $x \in [x_a, x_b]$ of problem (12)-(14) satisfies*

$$(x - c)^2 + (y(x) - d)^2 = r^2,$$

where c , d and r are determined by conditions (13)-(14).

Proof. If $y(x)$ is the solution to problem (12)-(14) then there exists a constant r , such that $y(x)$ is an ordinary extremal of the following functional [1]:

$$(15) \quad \int_{x_a}^{x_b} (y + r\sqrt{1 + y'^2}) dx.$$

The solution $y(x)$, $x \in [x_a, x_b]$ satisfies the Euler equation

$$\frac{d}{dx} \frac{ry'}{\sqrt{1 + y'^2}} = 1,$$

which after the integration becomes

$$\frac{ry'}{\sqrt{1 + y'^2}} = x - c,$$

where c is a constant of integration.

After squaring the last expression we obtain

$$(ry')^2 = (x - c)^2(1 + y'^2).$$

Since we are looking for a minimum of (12) then $\text{sign}(y') = \text{sign}(x - c)$. Therefore we can rewrite the last expression as

$$y' = \frac{x - c}{\sqrt{r^2 - (x - c)^2}},$$

which after integration becomes

$$y - d = \sqrt{r^2 - (x - c)^2},$$

and therefore

$$(x - c)^2 + (y(x) - d)^2 = r^2.$$

The proof of the proposition is complete.

Let us find a solution of (12)-(14) numerically. Figure 7 shows the corresponding AMPL model. Note that constraint (13) is relaxed to an inequality to make the problem convex:

$$\text{s.t. length: } \sum \{j \text{ in } 1..N\} \sqrt{1 + ydot[j]^2} * dx \leq 1;$$

Again, it can be shown by contradiction that the optimal solution of problem in Figure 7 with the relaxed “length” constraints satisfies these constraints as equalities. The solution of this problem is the semicircle (see Figure 8). Table 4 presents the statistics of the solver’s behavior.

Remark 3.2. For values of l within a certain range $l_1 \leq l \leq l_2$, the solution of problem (12)-(14) is an arc of a circle. However, if the length of a cable is big enough it is no longer possible to find a circle satisfying conditions (13)-(14). Then the solution of problem (12)-(14) may be found from a class of nonsmooth functions. Theoretical analysis corresponding to this case is complicated. It is simpler to find solutions numerically. For example, let us double the length of a chain:

$$\text{param } l := 2*d*2*\text{atan}(1).$$

<pre> param N := 200; param d := 5; param ax := 0; param ay := 0; param bx := d; param by := 0; param dx := (bx - ax)/N; param x {j in 0..N} := ax*(1-j/N) + bx*j/N; param l := d*2*atan(1); param g := 9.8; param m {0..N}, default 1; var y {0..N}; var ydot {j in 1..N}; minimize energy: sum {j in 1..N} dx*m[j]*g*(y[j]+y[j-1])/2; </pre>	<pre> s.t. eqn {j in 1..N}: y[j] = y[j-1] + dx*ydot[j]; s.t. length: sum {j in 1..N} sqrt(1+ydot[j]^2)*dx <= l; s.t. leftfixed: y[0] = ay; s.t. rightfixed: y[N] = by; let {j in 0..N} y[j] := (j/N)*by + (1-j/N)*ay; let {j in 1..N} ydot[j] := (y[j]-y[j-1])/dx; solve; </pre>
--	--


FIGURE 7. The AMPL model `fixedmass.mod`

FIGURE 8. Circle

The graph of the solution is shown in Figure 9 and the solver's behavior is shown in Table 5. Our presumption is that the curve AMNB in Figure 9 consists of circle segment MN and two vertical segments AM and NB.

Remark 3.3. The solution to problem (12)-(14) does not lead to a real suspension bridge model with uniformly distributed mass along the horizon. Actually, making an assumption of a mass distribution $m(x)$

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	0.22	0.98	6.43	57.79
pr. val =	-96.16397810	-96.19471719	-96.20546126	-96.2092295
dual val =	-96.16397809	-96.19471719	-96.20546125	-96.2092295
pr. inf =	1.7e-11	9.1e-12	2.7e-11	1.9e-11
dual inf =	2.6e-11	4.4e-12	2.1e-11	1.1e-11
steps	20	19	20	21

TABLE 4. Fixed uniformly distributed mass function along the horizontal axis

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	0.26	1.30	8.07	70.39
pr. val =	-286.7872655	-287.7015964	-288.1639555	-288.3974098
dual val =	-286.7872655	-287.7015964	-288.1639555	-288.3974098
pr. inf =	1.3e-12	1.1e-12	4.6e-12	2.5e-12
dual inf =	5.1e-12	1.5e-11	9.0e-12	5.8e-12
steps	23	25	25	26

TABLE 5. Long chain




FIGURE 9. Long circle




FIGURE 10. Fixed distribution of mass

in criteria (12) results in the appearance of horizontal external forces, which prevent redistribution of the chain mass along the horizon.

To clarify, let us consider the situation in Figure 10. Suppose we have fixed vertical rods with sliding frictionless rings of given mass and a weightless string, which goes through all the rings. This is exactly the case when the distribution of mass along the horizon is fixed. If the system rests in the state of minimal potential energy there will be normal horizontal forces coming from the rods. However the shape of the hanging chain is the result of just the gravity and tension forces.

3.2. Lessons. Problem (12)-(14) cannot handle the case when we want to find the shape of the a chain with a given distribution along the horizon $m(x)$ due to the following difficulty: we cannot use function $m(x)$ in the criteria (12) because it leads to the appearance of horizontal external forces and brings us to a wrong solution. The next subsection resolves this paradox when the chain is parameterized along itself.

3.3. Bridge model, discretization along the chain's length. In this section we consider the situation where the mass of the bridge is distributed along the horizon while the chain is parameterized along its length $(x(t), y(t))$, $t \in [0, l]$.

In the minimized integral criteria we cannot make assumptions on the mass distributions along the horizon (see Remark 3.3). Therefore we assume that mass distribution depends only on t . In the following, we consider again model (5)-(7) with a general mass distribution function $m(t)$

$$(5) \quad \min \int_0^l m(t)y(t)dt,$$

such that

$$(6) \quad \dot{x}^2 + \dot{y}^2 = 1,$$

and

$$(7) \quad x(0) = x_a, \quad y(0) = y_a, \quad x(l) = x_b, \quad y(l) = y_b.$$

For any distribution function $m(t)$ there is a solution to problem (5)-(7). However we are interested in the specific case where the mass is uniformly distributed along the horizon when the minimum is found. The following proposition proves that only the parabola can be such a solution.

Proposition 3. *Let $(x(t), y(t))$, $t \in [0, l]$ be the solution to problem (5)-(7) whose mass is uniformly distributed along the horizon, then*

$$(16) \quad y(x) = C_1 x^2 + C_2 x + C_3,$$

where the values of $C_1 > 0$, C_2 and C_3 are determined by conditions (6)-(7).

Proof. We can assume that $\dot{x} \neq 0$, $t \in [0, l]$, see (Proposition 3.1). Therefore if vector function $(x(t), y(t))$ is the solution of problems (5)-(7) then there exists a multiplier $\lambda = \lambda(t)$, $t \in [0, l]$ such that $(x(t), y(t))$ is an extremal without side condition for the functional

$$(17) \quad \int_0^l (m(t)y(t) + \lambda(t)(\dot{x}^2 + \dot{y}^2 - 1)) dt,$$

The Euler equations for problem (17) are

$$\begin{aligned} \frac{d}{dt} 2\lambda\dot{x} &= 0, \\ \frac{d}{dt} 2\lambda\dot{y} &= m(t). \end{aligned}$$

Integrating the first equation we obtain $2\lambda\dot{x} = C$. Keeping in mind that $\dot{x} > 0$, $t \in [0, l]$ we can express $\lambda = C/(2\dot{x})$ and substitute to the second equation, which becomes

$$(18) \quad C \frac{d}{dt} \frac{\dot{y}}{\dot{x}} = m(t).$$

Since we look for the solution with uniformly distributed mass along the horizon the following relation between $m(t)$ and $m(x)$ holds

$$(19) \quad m(t) = \frac{m(x)}{\sqrt{1 + y'^2}} = m(x)\dot{x} = m\dot{x},$$

where $m = M/|x_a - x_b|$ and M is the mass of whole chain. The second equality follows from (11). Therefore taking into account (10) and (19) we can rewrite (18) as follows

$$C y'' \dot{x} = m\dot{x}.$$

Since $\dot{x} > 0$ we have

$$y'' = \frac{C_1}{2}, \quad C_1 = \frac{2m}{C} > 0,$$

which after integrating two times becomes (16). This proves the proposition.

Remark 3.5. If instead of (19) one considers $m(t) = m$, $t \in [0, l]$ for ordinary differential equation (18), one can verify that $y(x)$ is a hyperbolic cosine (4). Therefore equations (18) and (6) can be treated as general necessary conditions for problem (5)-(7) for any given mass function $m(t)$.

Although there is no explicit optimization criteria in (6), (18), we still argue that the solution to this system can be obtained easily by using the solver. Let us solve this problem numerically for the case when the mass is distributed uniformly along the horizon

$$(20) \quad C \frac{d}{dt} \frac{\dot{y}}{\dot{x}} = m(t),$$

$$(21) \quad \dot{x}^2 + \dot{y}^2 = 1,$$

$$(22) \quad x(0) = x_a, \quad y(0) = y_a, \quad x(l) = x_b, \quad y(l) = y_b.$$

One possible way is to discretize (20), (21) without objective function subject to fixed end constraints (22). This approach can lead to numerical problems, since the problem becomes nonlinear and nonconvex. Instead we will try to "convexify" the problem to make it more manageable. Again we relax constraint (21) to inequality and introduce a linear objective function, which guarantees that a solution will satisfy constraint (21) as equality.

In order to simplify constraint (20) we integrate it

$$(23) \quad C \frac{\dot{y}}{\dot{x}} = \int_0^t m(\xi) d\xi - E = M(t) - E,$$

where a new variable $M(\bar{t}) = \int_0^{\bar{t}} m(\xi) d\xi$ is a total mass of a chain segment when $t \in [0, \bar{t}]$.

Discretizing $M(t)$ leads to the following

$$(24) \quad M_{t+1} = M_t + \int_t^{t+1} m(\xi) d\xi$$

If mass is distributed uniformly along the horizon then discretized (24) becomes

$$M[t+1] = M[t] + m*(x[t+1] - x[t]),$$

while discretized (23) yields

$$C*(y[t]-y[t-1]) + E*(x[t]-x[t-1]) = M[t]*(x[t]-x[t-1]);$$

where constants C and E will be treated as unknown variables. Figure 11 shows the complete discretized AMPL model and Table 6 presents the behavior of the solver.

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	0.30	0.68	1.63	3.4
pr. val =	-222.6832577	-222.6897622	-222.6913882	-222.6917948
dual val =	-222.6832577	-222.6897622	-222.6913882	-222.6917948
pr. inf =	1.2e-12	4.1e-13	6.3e-13	3.4e-13
dual inf =	4.1e-13	8.5e-14	5.0e-14	5.7e-15
steps	23	25	28	30

TABLE 6. Parabolic chain

Remark 3.6. To obtain the solution with any distribution function $m(t)$ it is enough to change the "mint" constraint. For example to obtain the catenary solution ($m(t) = m, t \in [0, l]$) it is enough to replace the "mint" constraint with the following:

<pre> param N := 200; param d:= 5; param ax := 0; param ay := 0; param bx := d; param by := 0; param l := 2*d; param g := 9.8; param m {1..N}, default 1; var x {0..N}; var y {0..N}; var C; var E; var M {0..N}; minimize energy: (1/N)*sum {j in 1..N-1} m[j]*g*y[j]; s.t. xleftfixed: x[0] = ax; s.t. yleftfixed: y[0] = ay; s.t. xrightfixed: x[N] = bx; s.t. yrightfixed: y[N] = by; </pre>	<pre> s.t. link {j in 1..N}: (x[j]-x[j-1])^2+(y[j]-y[j-1])^2<= (1/N)^2; s.t. eqlb {j in 1..N}: M[j]*(x[j]-x[j-1]) = C*(y[j]-y[j-1]) + E*(x[j]-x[j-1]); s.t. mint {j in 0..N-1}: M[j+1] = M[j] + m[j+1]*(x[j+1]-x[j]); s.t. mifixed: M[0] = 0; let {j in 0..N} x[j] := (j/N)*b_x + (1-j/N)*a_x; let {j in 0..N} y[j] := (j/N)*b_y + (1-j/N)*a_y; solve; </pre>
---	---

FIGURE 11. The AMPL model parab.mod

s.t. mint {j in 0..N-1}: M[j+1] = M[j] + m[j+1].

Table 7 shows the solver's behavior while Figure 12 shows shapes of the parabolic and catenary chains.

	$N = 100$	$N = 200$	$N = 400$	$N = 800$
time	0.34	1.10	3.74	13.18
pr. val =	-223.2381140	-223.2440837	-223.2455761	-223.2459492
dual val =	-223.2381140	-223.2440837	-223.2455760	-223.2459492
pr. inf =	2.2e-12	4.5e-13	1.5e-12	1.6e-13
dual inf =	2.9e-11	2.7e-12	1.9e-12	6.7e-12
steps	32	51	82	149

TABLE 7. Catenary chain

4. CONCLUSION.

Analyzing the hanging chain model we have come to several conclusions. First, discretizing the model along the chain gives more flexibility in modeling various cases. In particular, we were able to model the case when the mass function is uniformly distributed along the bridge span. Also, we explained the difficulties that prevented the modeling of this situation when the problem is discretized along the horizon.

Second, when the model is discretized along the chain, it is significantly simpler to solve the corresponding optimization problem using a nonlinear optimizer. In particular, the catenary problem with known mass function $m(t)$ becomes a convex optimization problem. As a result, for LOQO, it takes seconds to find the solution even for a large number of discretization points.




FIGURE 12. Parabolic and catenary chains

Third, after solving a variety of models we are convinced in the importance of a proper modeling. Problems can often be reformulated to create a simpler optimization problem and thereby reduce significantly the solution time.

Finally, we have chosen LOQO as a solver for this paper. Those who are interested in the behavior of different solvers for this class of problems can refer to [3, 4], where one can find comparative studies for different solvers, and also to NEOS website <http://www-neos.mcs.anl.gov/>, where one can use the best optimization software for problems formulated in AMPL.

Acknowledgements. The first author would like to thank Vladimir Aksyuk and Michael Libman for fruitful discussions of the paper. He is also grateful to Galina Spivak for technical assistance.

REFERENCES

- [1] N.I. Akhiezer, *The Calculus of Variations*, New York, Blaisdell Pub. Co., 1962. 3, 6, 10
- [2] I.M. Gelfand, S.V. Fomin, *Calculus of Variations*, Englewood Cliffs, NJ, Prentice-Hall, Inc, 1963. 3
- [3] A.S. Bondarenko, D.M. Bortz, J.J. More, *COPS: Constrained optimization problems*, Mathematics and Computer Science Division, Argonne National Laboratory, <http://www-unix.mcs.anl.gov/more/cops> 17

- [4] A.S. Bondarenko, D.M. Bortz, J.J. More, *COPS: Large-scale nonlinearly constrained optimization problems*, Mathematics and Computer Science Division, Argonne National Laboratory, Technical Report ANL/MCS-TM-237, October 1999 17
- [5] R. Fourer, D.M. Gay, and B.W. Kernigan, *AMPL : A modeling Language for Mathematical Programming*, Scientific Press, 1993. 2
- [6] R. J. Vanderbei, D.F. Shanno, *An interior-point algorithm for nonconvex nonlinear programming*, COAP 13, 1999, 231-252. 2
- [7] R. J. Vanderbei, *LOQO: An interior point code for quadratic programming*, Optimization Methods and Software, 12, 1999, 451-484. 2
- [8] R. J. Vanderbei, *LOQO user's manual - version 3.10*, Optimization methods and software, 12, 1999, 485-514. 2
- [9] R. J. Vanderbei, *Case Studies In Trajectory Optimization: Trains, Plains, and Other Pastimes*, Technical Report ORFE-00-3, Department of Operations Research and Financial Engineering. To appear in *Optimization and Engineering*. 1, 3

PRINCETON UNIVERSITY, PRINCETON, NJ